
Getting Started
    

There are a few steps needed to get started with the Ecspanse ECS framework. This guide covers the process of setting up the environment and creating a working application.

  
    
  
  Create a new Elixir/Phoenix project


This guide we will use a simple supervised Elixir application:
mix new demo --sup


  
    
  
  Installation


The first step is adding the :ecspanse library to the project dependencies. This is done by adding the following line to the mix.exs file:
def deps do
  [
    {:ecspanse, "~> 0.9.0"}
  ]
end

  
    
  
  Setup


The subsequent stage involves configuring Ecspanse via use Ecspanse. In this guide, the primary Demo module will accommodate the Ecspanse setup:
defmodule Demo do
  use Ecspanse

  @impl Ecspanse
  def setup(data) do
    data
  end
end
The Ecspanse.setup/1 callback is mandatory. It will be used later on to schedule the application systems.

  
    
  
  Starting the Ecspanse server


The module invoking use Ecspanse needs to be added to the aplication supervision tree.
defmodule Demo.Application do
  @moduledoc false

  use Application

  @impl true
  def start(_type, _args) do
    children = [
      Demo
    ]

    opts = [strategy: :one_for_one, name: Demo.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
The Ecspanse server loop will start together with the application:
iex -S mix



  

    
Tutorial
    

The objective of this tutorial is to develop a basic game utilizing the Ecspanse framework. The game is the initial stage of an RPG game with minimal features. The implementation focuses on the on the game logic and it does not have a UI. All the user interactions and minimal display will be done via Livebook integration.

  
    
  
  Story


The story of the game is simple:
	the game features a single character (Hero).
	the hero has energy. It starts with 50 energy points and it can have a maximum of 100 energy points. The energy points are used to perform actions. Every 3 seconds the hero restores 1 energy point.
	the hero can move in four directions in a tiles-like manner, without actually implementing a tile system. For example, if the hero moves right it will transition from (0,0) to (1,0) and so on. Each move costs 1 energy point.
	on each move the hero has a chance to find resources: gold or gems. Resources are not inventory items, but tradeable items. The hero can trade resources for inventory items.
	the hero starts with some items in their inventory: 2 potions and one pair of boots.
	the hero can purchase a map with 2 gold and a compass with 3 gold and 2 gems.

This setup enables us to delve into fundamental concepts of ECS in general and Ecspanse in particular:
	creating new entities from components
	querying for components
	interacting with the system via events
	scheduling systems to perform actions
	managing entities relationships
	different ways to approach collections of entities or components
	using time-based systems and events



  
    
  
  Spawning the Hero


The goal of this chapter is to spawn the hero entity with its components on game startup.

  
    
  
  Ecspanse Concepts 1


	creating components
	using commands to spawn entities
	creating and scheduling systems
	querying components



  
    
  
  Adding the Components


The hero entity will have for now the following components:
defmodule Demo.Components.Hero do
  use Ecspanse.Component, state: [name: "Hero"]
end

defmodule Demo.Components.Energy do
  use Ecspanse.Component, state: [current: 50, max: 100]
end

defmodule Demo.Components.Position do
  use Ecspanse.Component, state: [x: 0, y: 0]
end

The Hero component holds generic information about the hero. The Energy holds the current and maximum energy points. The Position component holds the current position of the hero as horizontal and vertical coordinates.
Under the hood, the components are structs, with some metadata added by the library.
The following options are available when defining a component:
	:state - the fields and the initial state of the component. It should be a list or a keyword list.
	:tags - a list of atoms that can be used to tag the component. Tags are an alternate way of querying components.


  
    
  
  The Hero Entity Spec


While this is not mandatory, we extracted the hero entity spec composition in the Demo.Entities.Hero module.
defmodule Demo.Entities.Hero do
  alias Demo.Components

  @spec new() :: Ecspanse.Entity.entity_spec()
  def new do
    {Ecspanse.Entity,
     components: [
       Components.Hero,
       Components.Energy,
       Components.Position
     ]}
  end
end
The new/0 function does not have any effect. It just prepares the entity spec (of type Ecspanse.Entity.entity_spec/0) to be spawned.

  
    
  
  The Spawn Hero System


The system that spawns the hero is a simple one. It is scheduled to run once, when the game starts.
defmodule Demo.Systems.SpawnHero do
  use Ecspanse.System

  @impl true
  def run(_frame) do
    %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Hero.new())
  end
end
The system must implement the Ecspanse.System.WithoutEventSubscriptions.run/1 or Ecspanse.System.WithEventSubscriptions.run/2 callback. In this case, it is a generic system, not subscribing to any events, so we will use run/1. The callback receives the Ecspanse.Frame.t/0 as argument.
Operations that involve the creation of entities or components are done via the functions in the Ecspanse.Command module. The commands cannot be executed outside of a system.

  
    
  
  Scheduling the Spawn Hero System


It is now time to schedule the newly created system as a startup system. This is done by updating the Ecspanse.setup/1 function in the Demo module. We already created this function in the Getting Started guide.
defmodule Demo do
  use Ecspanse

  alias Demo.Systems

  @impl Ecspanse
  def setup(data) do
    data
    |> Ecspanse.add_startup_system(Systems.SpawnHero)
  end
end
If we start the application now, the hero entity will be spawned.

  
    
  
  Querying for the Hero


Next, we will incorporate some helper functions that will prove useful in the next chapters.
defmodule Demo.Entities.Hero do
  alias Demo.Components

  #...

  def fetch do
    Ecspanse.Query.select({Ecspanse.Entity}, with: [Components.Hero])
    |> Ecspanse.Query.one()
    |> case do
      {%Ecspanse.Entity{} = entity} -> {:ok, entity}
      _ -> {:error, :not_found}
    end
  end
end
In the Demo.Entities.Hero module we added the fetch/0 function. This function uses the Ecspanse.Query module to select the hero entity. The Ecspanse.Query.select/2 function is the most flexible way to query for entities and components and it will be used many times in this tutorial. In the current context, the query can be interpreted as:
	select tuples with a single element, Entity -> this queries the %Ecspanse.Entity{} struct itself. When we want to return the entity as part of a more complex select query, it needs to be in the first position of the tuple.
	that has the Demo.Components.Hero component attached to it.
	return just one record -> this would return the selected entity tuple if found, or otherwise nil. It is important to note that, if many records match the query, it will raise an error. For such cases, the Ecspanse.Query.stream/1 should be used and it will return a stream of results.

We can actually test the function in the iex console after starting the server:
iex(1)> Demo.Entities.Hero.fetch()
{:ok, %Ecspanse.Entity{id: "e950bf44-16d5-46b5-bd21-85aabae50ce8"}}
For the other helper function, we will create an API module. Again, this is not part of the library, but it provides an easy way to interact with the game, no matter what front end we will use.
defmodule Demo.API do
  @spec fetch_hero_details() :: {:ok, map()} | {:error, :not_found}
  def fetch_hero_details do
    Ecspanse.Query.select(
      {Ecspanse.Entity, Demo.Components.Hero, Demo.Components.Energy, Demo.Components.Position}
    )
    |> Ecspanse.Query.one()
    |> case do
      {hero_entity, hero, energy, position} ->
        {:ok, %{name: hero.name, energy: energy.current, max_energy: energy.max, pos_x: position.x, pos_y: position.y}}
      _ ->
        {:error, :not_found}
    end
  end
end
This function, returns a map with the hero details we have implemented so far. This time, the select/2 function is used to select multiple components. The query can be interpreted as:
	select tuples with three elements, Demo.Components.Hero, Demo.Components.Energy, Demo.Components.Position only from entities that have all three components attached to them.
	return just one record -> this will return a tuple with the three components structs if found, or nil otherwise.

We can test the function in the iex console after starting the server:
iex(2)> Demo.API.fetch_hero_details()
%{name: "Hero", energy: 50, max_energy: 100, pos_x: 0, pos_y: 0}


  
    
  
  Hero Movement


The goal of this chapter is to implement the hero movement. The hero will be able to move in the four directions: up, down, left and right.

	receiving external input through events
	implementing and scheduling async systems
	locking components for parallel operations
	implementing systems that subscribe to events
	updating components with commands



  
    
  
  The Move Event


Ecspanse receives external input through events. Let's implement the move event.
defmodule Demo.Events.MoveHero do
  use Ecspanse.Event, fields: [:direction]
end
Similar to the components, the events are structs under the hood. The fields and their default values are defined with the :fields option.
We can also expose the event in the API module:
defmodule Demo.API do
  #...
  @spec move_hero(direction :: :up | :down | :left | :right) :: :ok
  def move_hero(direction) do
    Ecspanse.event({Demo.Events.MoveHero, direction: direction})
  end
end
We create another event that will be emitted when the hero actually moved to handle various side effects.
defmodule Demo.Events.HeroMoved do
  use Ecspanse.Event
end

  
    
  
  The Move System


The role of the move system is to listen to move events, then check if the hero has enough energy to move and if so, update the hero position and adjust the energy. We want this system to run asynchronously.
defmodule Demo.Systems.MoveHero do
  use Ecspanse.System,
    lock_components: [Demo.Components.Position, Demo.Components.Energy],
    event_subscriptions: [Demo.Events.MoveHero]

  alias Demo.Components

  @impl true
  def run(%Demo.Events.MoveHero{direction: direction}, _frame) do
    components =
      Ecspanse.Query.select({Components.Position, Components.Energy}, with: [Components.Hero])
      |> Ecspanse.Query.one()

    with {position, energy} <- components,
         :ok <- validate_enough_energy_to_move(energy) do
      Ecspanse.Command.update_components!([
        {energy, current: energy.current - 1},
        {position, update_coordinates(position, direction)}
      ])
      Ecspanse.event(Demo.Events.HeroMoved)
    end
  end

  defp validate_enough_energy_to_move(%Components.Energy{current: current_energy}) do
    if current_energy >= 1 do
      :ok
    else
      {:error, :not_enough_energy}
    end
  end

  defp update_coordinates(%Components.Position{x: x, y: y}, direction) do
    case direction do
      :up -> [x: x, y: y + 1]
      :down -> [x: x, y: y - 1]
      :left -> [x: x - 1, y: y]
      :right -> [x: x + 1, y: y]
      _ -> [x: x, y: y]
    end
  end
end
Component Locking
We said that the move system will run asynchronously. This means that it will run in parallel with the other systems. The lock_components option is used to specify the components that will be locked by the system. That means that no other systems that lock at least one of the locked components will run in the same parallel batch as the MoveHero system. In our case, we want to lock the Demo.Components.Position and Demo.Components.Energy components. This is because we want to update the hero position and energy, and we need to avoid race conditions.
The commands will check if the system is async and will raise an error if we try to update, insert or delete components that are not locked. For extra safety we can also lock components for which we don't update the state, but we read and depend on it.
Event Subscriptions
Not all systems are required to run every single frame. The MoveHero system is useful only when a MoveHero event is received. The event_subscriptions option is used to specify the events that the system is interested in. The system will run only when at least one of the subscribed events is received.
The systems that have events subscriptions need to implement the run/2 callback. The first argument is the event that triggered the system. The second argument is the current frame.
Updating the Components
It is a good practice to make sure that actions can be performed before committing any updates. Reverting the changes would be more difficult and inefficient.
In our case, we want to make sure that the hero has enough energy before updating the Energy and Position components state.

  
    
  
  Scheduling the Move System


We use add_system/2 to schedule the MoveHero system to run asynchronously.
defmodule Demo do
  #...
  def setup(data) do
    data
    #...
    |> Ecspanse.add_system(Systems.MoveHero)
  end
end
We can try the hero movement in the iex console:
iex(1)> Demo.API.fetch_hero_details()
%{name: "Hero", energy: 50, max_energy: 100, pos_x: 0, pos_y: 0}
iex(2)> Demo.API.move_hero(:up)
:ok
iex(3)> Demo.API.move_hero(:right)
:ok
iex(4)> Demo.API.fetch_hero_details()
%{name: "Hero", energy: 48, max_energy: 100, pos_x: 1, pos_y: 1}


  
    
  
  Energy Regeneration


The goal of this chapter is to implement the energy regeneration. The hero will restore 1 point of energy every 3 seconds.

  
    
  
  Ecspanse Concepts 3


	using the timer to schedule events at precise intervals
	using built-in component and event templates
	ordering async systems
	conditionally running systems
	new ways of querying entities and components



  
    
  
  The Energy Timer Component


Timer-based components use the provided Ecspanse.Template.Component.Timer component template. The timer component is a special component that is used to schedule events at precise intervals. The timer component template exposes the following fields:
	:duration - the countdown duration in milliseconds
	:time - the current countdown time in milliseconds
	:event - the event that will be triggered when the countdown reaches 0
	:mode - :repeat | :once | :temporary - decides the timer behavior after the countdown reaches 0.

	:paused - boolean - can be used to pause the timer

We will discuss more about template components in the next chapter.
defmodule Demo.Components.EnergyTimer do
  use Ecspanse.Template.Component.Timer,
    state: [duration: 3000, time: 3000, event: Demo.Events.EnergyTimerFinished, mode: :repeat]
end
We add the new component to the Hero entity:
defmodule Demo.Entities.Hero do
  #...
  @spec new() :: Ecspanse.Entity.entity_spec()
  def new do
    {Ecspanse.Entity,
    components: [
      Components.Hero,
      Components.Energy,
      Components.Position,
      Components.EnergyTimer
    ]}
  end
  #...
end

  
    
  
  The Energy Timer Finished Event


Timer-based events use the provided Ecspanse.Template.Event.Timer event template. The timer event template exposes the following fields:
	entity_id - the id of the entity that owns the timer component

defmodule Demo.Events.EnergyTimerFinished do
  use Ecspanse.Template.Event.Timer
end
This event will be automatically triggered when the EnergyTimer component duration reaches 0.

  
    
  
  The Energy Restore System


defmodule Demo.Systems.RestoreEnergy do
  use Ecspanse.System,
    lock_components: [Demo.Components.Energy],
    event_subscriptions: [Demo.Events.EnergyTimerFinished]

  @impl true
  def run(%Demo.Events.EnergyTimerFinished{entity_id: entity_id}, _frame) do
    with {:ok, entity} <- Ecspanse.Query.fetch_entity(entity_id),
         {:ok, energy} <- Ecspanse.Query.fetch_component(entity, Demo.Components.Energy) do
      Ecspanse.Command.update_component!(energy, current: energy.current + 1)
    end
  end
end
The system locks the Energy component to update its state. And subscribes to the EnergyTimerFinished event because it is interested only in that timer event.
The system adds 1 point to the current energy. You are right to wonder why don't we first check the max energy cap. We will clarify this in the next section.
This system also introduces new ways of querying entities and components: Ecspanse.Query.fetch_entity/1 and Ecspanse.Query.fetch_component/2.
TIP
The following functions would produce the same results:
Ecspanse.Query.fetch_component(entity, Demo.Components.Energy)
#and
Demo.Components.Energy.fetch(entity)

  
    
  
  Rescheduling the Systems Execution


It is time to re-write the Demo module:
defmodule Demo do
  use Ecspanse

  alias Demo.Systems

  @impl Ecspanse
  def setup(data) do
    data
    |> Ecspanse.add_startup_system(Systems.SpawnHero)
    |> Ecspanse.add_system(Systems.RestoreEnergy, run_if: [{__MODULE__, :energy_not_max?}])
    |> Ecspanse.add_system(Systems.MoveHero, run_after: [Systems.RestoreEnergy])
    |> Ecspanse.add_frame_end_system(Ecspanse.System.Timer)
  end

  def energy_not_max? do
    Ecspanse.Query.select({Demo.Components.Energy}, with: [Demo.Components.Hero])
    |> Ecspanse.Query.one()
    |> case do
      {%Demo.Components.Energy{current: current, max: max}} ->
        current < max

      _ ->
        false
    end
  end
end

The Conditional System Execution
By using the :run_if option, the RestoreEnergy system will run only if the current energy is below the max energy. The energy_not_max?/0 function must always return a boolean value. Please note, this is not an efficient implementation. The energy_not_max?/0 function will be called every frame. If the check would happen in the RestoreEnergy system, it would run only once every 3 seconds. But we took the opportunity to exemplify conditionally running systems.
The System Execution Order
By using the :run_after option, the MoveHero system will run after the RestoreEnergy system. Both are async systems, but even the async systems run in batches, not all at once. The batches are scheduled depending on the locked components and the specified order of execution of the systems.
Note
It does not matter if the RestoreEnergy system actually runs this turn.
The MoveHero will still run if receiving the MoveHero event.
The :run_after option just guarantees that if both systems are running,
the MoveHero will run after the RestoreEnergy.

Scheduling the Built-in Timer System
Once we start using timer-based components, the built-in Ecspanse.System.Timer system must be scheduled to run synchronously at the beginning or at the end of every frame. It will update all the timer-based components and trigger the timer events.


  
    
  
  Finding Resources


The goal of this chapter is to implement the resource gathering. With each move, the hero has a chance to find gold or gems.

  
    
  
  Ecspanse Concepts 4


	using tags to manage collections of components
	using advanced component specs
	using component templates
	using the auto-emitted component updated events



  
    
  
  Creating the Resource Template Component


Template components are used to define the structure for related components. It is a guarantee that certain components will have certain fields in their state.
Together with tags, this is a powerful way to achieve polymorphism in components.
defmodule Demo.Components.Resource do
  use Ecspanse.Template.Component, state: [:id, :name, amount: 0], tags: [:resource]

  @impl true
  def validate(state) do
    with :ok <- validate_integer_amount(state[:amount]),
         :ok <- validate_positive_amount(state[:amount]) do
      :ok
    end
  end

  defp validate_integer_amount(amount) do
    if is_integer(amount) do
      :ok
    else
      {:error, "#{inspect(amount)} must be an integer"}
    end
  end

  defp validate_positive_amount(amount) do
    if amount >= 0 do
      :ok
    else
      {:error, "#{inspect(amount)} must be positive"}
    end
  end
end
Please note that the template Ecspanse.Template.Component.validate/1 callback is optional. It runs only at compile time and it takes the list of state fields as argument.

  
    
  
  Creating the Resource Components


defmodule Demo.Components.Gems do
  use Demo.Components.Resource,
    state: [id: :gems, name: "Gems", amount: 0], tags: [:resource]
end

defmodule Demo.Components.Gold do
  use Demo.Components.Resource,
    state: [id: :gold, name: "Gold", amount: 0], tags: [:resource]
end
As you can observe, the two components are invoking the newly defined template with use Demo.Components.Resource instead of use Ecspanse.Component.
Another new concept introduced both here and in the template definition is the :tags option. It is a list of atoms that can be used to group and query components. The resource components can now be used as a resource store for the user, but they can also be used to represent the cost of various items. We will handle the second use case in the next chapters.
For such cases, it is important to use a standardized approach, a perfect use-case for templates. E.g., all the resource components should have the same state fields.

  
    
  
  Adding the Resources Components to the Hero Entity


defmodule Demo.Entities.Hero do
  #...

  def new do
    {Ecspanse.Entity,
     components: [
        #...
       {Components.Gold, [], [:available]},
       {Components.Gems, [], [:available]}
     ]}
  end
  #...
end
We use the Ecspanse.Component.component_spec/0 type to specify the component spec. The first element of the tuple is the component module, the second element is the initial state of the component, and the third element is a list of tags.
The initial state of the component can be changed at runtime like {Components.Gold, [amount: 5], [:available]}. Also, new tags can be added at the time of the component creation. They will be appended to the list defined in the component module.
Runtime tag setting enables the reusability of components in various scenarios. It's important to note that changing a component's tags after it has been created is not supported.

  
    
  
  Storing Found Resources


Our new MaybeFindResources system subscribes to Demo.Events.HeroMoved emitted by the MoveHero system. Then it randomly decides if the current position contains resources, and the type of resource.
defmodule Demo.Systems.MaybeFindResources do
  use Ecspanse.System,
    lock_components: [Demo.Components.Gems, Demo.Components.Gold],
    event_subscriptions: [Demo.Events.HeroMoved]

  alias Demo.Components

  @impl true
  def run(%Demo.Events.HeroMoved{}, _frame) do
    with true <- found_resource?(),
         resource_module <- pick_resource(),
         {:ok, hero_entity} <- Demo.Entities.Hero.fetch(),
         {:ok, resource} <- Ecspanse.Query.fetch_component(hero_entity, resource_module) do
      Ecspanse.Command.update_component!(resource, amount: resource.amount + 1)
    end
  end

  def run(_event, _frame), do: :ok

  defp found_resource?, do: Enum.random([true, false])
  defp pick_resource, do: Enum.random([Components.Gems, Components.Gold])
end
Here we take advantage of the standardized resource approach, so the system would update the resource amount without caring about the actual resource type.
Then we add the new system to the setup:
defmodule Demo do
  use Ecspanse
  # ...
  def setup(data) do
    data
    # ...
    |> Ecspanse.add_system(Systems.MoveHero, run_after: [Systems.RestoreEnergy])
    |> Ecspanse.add_system(Systems.MaybeFindResources)
    |> Ecspanse.add_frame_end_system(Ecspanse.System.Timer)
  end
end
The last step of the current section is to expose the resources in the fetch_hero_details/0 function in the Demo.API module.
  defp list_hero_resources(hero_entity) do
    hero_entity
    |> Ecspanse.Query.list_tagged_components_for_entity([:resource, :available])
    |> Enum.map(&%{name: &1.name, amount: &1.amount})
  end
Here we use the Ecspanse.Query.list_tagged_components_for_entity/2 function to get all the components tagged with :resource and :available for the hero entity.
The map returned by fetch_hero_details/0 function should be updated with the new resources field:
  %{
    # ...
    pos_y: position.y,
    resources: list_hero_resources(hero_entity),
  }
Starting the application and moving the hero around will now start to accumulate resources:
iex(14)> Demo.API.fetch_hero_details
%{
  name: "Hero",
  resources: [%{name: "Gems", amount: 2}, %{name: "Gold", amount: 5}],
  energy: 56,
  max_energy: 100,
  pos_x: -3,
  pos_y: -5
}
--

  
    
  
  Market and Inventory Items


The goal of this chapter is to implement inventory items and a market. The hero can buy items from the market with resources and store them in the inventory.

	using relationships to manage collections of entities
	querying components within entities relationships



  
    
  
  Inventory Items Components and Entities Specs


We start by defining the inventory items and the market components:
defmodule Demo.Components.Market do
  use Ecspanse.Component
end

defmodule Demo.Components.Boots do
  use Ecspanse.Component, state: [name: "Boots"], tags: [:inventory]
end

defmodule Demo.Components.Compass do
  use Ecspanse.Component, state: [name: "Compass"], tags: [:inventory]
end

defmodule Demo.Components.Map do
  use Ecspanse.Component, state: [name: "Map"], tags: [:inventory]
end

defmodule Demo.Components.Potion do
  use Ecspanse.Component, state: [name: "Potion"], tags: [:inventory]
end
The inventory items, however are more complex than this. They cost resources, and in the future they may have various attributes impacting the hero's abilities. So the items will be entities of their own. We will create a new Entities.Inventory module to manage the inventory items specs.
defmodule Demo.Entities.Inventory do
  alias Demo.Components

  @spec new_boots() :: Ecspanse.Entity.entity_spec()
  def new_boots do
    {Ecspanse.Entity, components: [Components.Boots, {Components.Gold, [amount: 3], [:cost]}]}
  end

  @spec new_compass() :: Ecspanse.Entity.entity_spec()
  def new_compass do
    {Ecspanse.Entity,
     components: [
       Components.Compass,
       {Components.Gold, [amount: 3], [:cost]},
       {Components.Gems, [amount: 2], [:cost]}
     ]}
  end

  @spec new_map() :: Ecspanse.Entity.entity_spec()
  def new_map do
    {Ecspanse.Entity, components: [Components.Map, {Components.Gold, [amount: 2], [:cost]}]}
  end

  @spec new_potion() :: Ecspanse.Entity.entity_spec()
  def new_potion do
    {Ecspanse.Entity, components: [Components.Potion, {Components.Gold, [amount: 1], [:cost]}]}
  end

  @spec list_inventory_components(Ecspanse.Entity.t()) :: [component :: struct()]
  def list_inventory_components(parent) do
    Ecspanse.Query.list_tagged_components_for_children(parent, [:inventory])
  end
end
Each item is defined together with their cost in resources. Please note that now we are using the :cost tag to mark the cost resource components.
The list_inventory_components/1 function is used to list all the inventory items for a given parent entity. We will see what this means in the next section.

  
    
  
  Inventory Items as Children Entities


Let's start by updating the existing SpawnHero system.
defmodule Demo.Systems.SpawnHero do
  use Ecspanse.System

  @impl true
  def run(_frame) do
    hero_entity = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Hero.new())
    potion_entity_1 = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Inventory.new_potion())
    potion_entity_2 = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Inventory.new_potion())
    boots_entity = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Inventory.new_boots())

    Ecspanse.Command.add_children!([ {hero_entity, [potion_entity_1, potion_entity_2, boots_entity]} ])
  end
end
The hero starts the journey with two potions and a pair of boots. We use the Ecspanse.Command.add_children!/1 function to add the inventory items as children of the hero entity. This way we can build complex entities by composing smaller entities.
The Ecspanse library provides many helper functions to query and change entities relations.
The next step is to create a new system that spawns a market entity that holds more items.
defmodule Demo.Systems.SpawnMarket do
  use Ecspanse.System

  @impl true
  def run(_frame) do
    compass_entity = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Inventory.new_compass())
    map_entity = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Inventory.new_map())

    Ecspanse.Command.spawn_entity!({ Ecspanse.Entity,
      components: [Demo.Components.Market], children: [compass_entity, map_entity]
    })
  end
end
This shows another way to spawn an entity with children already attached. The new system needs to be added to the setup/1 as startup system:
#...
|> Ecspanse.add_startup_system(Systems.SpawnMarket)
#...
One last thing we can do in this chapter is to add new functions to our API:
defmodule Demo.API do
  #...
  defp list_hero_inventory(hero_entity) do
    hero_entity
    |> Demo.Entities.Inventory.list_inventory_components()
    |> Enum.map(&%{name: &1.name})
  end

  @spec fetch_market_items() :: {:ok, list(map())} | {:error, :not_found}
  def fetch_market_items do
    Ecspanse.Query.select({Ecspanse.Entity}, with: [Demo.Components.Market])
    |> Ecspanse.Query.one()
    |> case do
      {market_entity} -> list_market_items(market_entity)
      _ -> {:error, :not_found}
    end
  end

  defp list_market_items(market_entity) do
    market_entity
    |> Demo.Entities.Inventory.list_inventory_components()
    |> Enum.map(fn item_component ->
      item_entity = Ecspanse.Query.get_component_entity(item_component)

      %{entity_id: item_entity.id, name: item_component.name, cost: item_cost(item_entity)}
    end)
  end

  defp item_cost(item_entity) do
    item_entity
    |> Ecspanse.Query.list_tagged_components_for_entity([:resource, :cost])
    |> Enum.map(&%{name: &1.name, amount: &1.amount})
  end
end
The map returned by fetch_hero_details/0 function should be updated with the new inventory field:
  %{
    # ...
    pos_y: position.y,
    resources: list_hero_resources(hero_entity),
    inventory: list_hero_inventory(hero_entity)
  }
We will now display the hero's inventory and the market items with their respective prices.
We can test the new functions in the iex console:
iex(1)> Demo.API.fetch_hero_details()
%{
  name: "Hero",
  resources: [%{name: "Gems", amount: 0}, %{name: "Gold", amount: 0}],
  inventory: [%{name: "Boots"}, %{name: "Potion"}, %{name: "Potion"}],
  energy: 60,
  max_energy: 100,
  pos_x: 0,
  pos_y: 0
}

iex(1)> Demo.API.fetch_market_items()
[
  %{
    name: "Map",
    entity_id: "361c00ba-4dd3-4be8-b171-00e99c0b8ef7",
    cost: [%{name: "Gold", amount: 2}]
  },
  %{
    name: "Compass",
    entity_id: "b027fd01-d4fe-4ac1-9736-6b6f8c58fbd1",
    cost: [%{name: "Gold", amount: 3}, %{name: "Gems", amount: 2}]
  }
]


  
    
  
  Purchasing Items from the Market


The goal of this chapter is to allow the hero to purchase items from the market using resources.

	in-depth entity relationships



  
    
  
  Purchasing Items Event


The event that triggers an item purchase is very simple:
defmodule Demo.Events.PurchaseMarketItem do
  use Ecspanse.Event, fields: [:item_entity_id]
end
It stores only the ID of the entity of the item being purchased.
On the other hand, the system is a bit more complex.

  
    
  
  Purchasing Items System


Let's start with the code:
defmodule Demo.Systems.PurchaseMarketItem do
  use Ecspanse.System, event_subscriptions: [Demo.Events.PurchaseMarketItem]

  @impl true
  def run(%Demo.Events.PurchaseMarketItem{item_entity_id: item_entity_id}, _frame) do
    with {:ok, item_entity} <- Ecspanse.Query.fetch_entity(item_entity_id),
         {:ok, market_entity} <- fetch_market_entity(),
         {:ok, hero_entity} <- Demo.Entities.Hero.fetch(),
         true <- Ecspanse.Query.is_child_of?(parent: market_entity, child: item_entity),
         hero_available_resources_components =
           Ecspanse.Query.list_tagged_components_for_entity(hero_entity, [:resource, :available]),
         item_cost_components =
           Ecspanse.Query.list_tagged_components_for_entity(item_entity, [:resource, :cost]),
         true <- has_enough_resources?(hero_available_resources_components, item_cost_components) do
      spend_resources(hero_available_resources_components, item_cost_components)
      Ecspanse.Command.remove_child!(market_entity, item_entity)
      Ecspanse.Command.add_child!(hero_entity, item_entity)
    end
  end

  defp fetch_market_entity do
    Ecspanse.Query.select({Ecspanse.Entity}, with: [Demo.Components.Market])
    |> Ecspanse.Query.one()
    |> case do
      {market_entity} -> {:ok, market_entity}
      _ -> {:error, :not_found}
    end
  end

  defp has_enough_resources?(available_resources, cost_resources) do
    Enum.all?(cost_resources, fn cost_resource ->
      Enum.any?(available_resources, fn available_resource ->
        available_resource.id == cost_resource.id &&
          available_resource.amount >= cost_resource.amount
      end)
    end)
  end

  defp spend_resources(available_resources, cost_resources) do
    Enum.each(cost_resources, fn cost_resource ->
      available_resource =
        Enum.find(available_resources, fn available_resource ->
          available_resource.id == cost_resource.id
        end)

      Ecspanse.Command.update_component!(available_resource,
        amount: available_resource.amount - cost_resource.amount
      )
    end)
  end
end
This system modifies many components, so one option is to make it synchronous. This way we don't have to individually lock each modified component. Later on it can be refactored into async if needed.
Before committing any component state changes it is important to perform all the required validations.
Validate Entities Exist
First of all, we want to make sure that the affected entities still exist. We use the Ecspanse.Query.fetch_entity/1 function to validate that the item entity exists. For the market entity we implement a custom query, while for the hero, we use the helper function we created earlier.
Validate Relationships
We need to make sure that the item we want to purchase is still available in the market. In a multiplayer game scenario this would avoid race conditions where two players purchase the same item in the same time. We use the Ecspanse.Query.is_child_of?/1 function to validate that the item is still a child of the market.
Validate Resources
Before the purchase is made, we need to make sure that the hero has enough resources to buy the item. Again, the tags prove useful. They allow us to query the same components from different entities and compare them.
Spending the Resources
Once all the validations are done, the resources can be spent. We iterate through the costs, then reduce the amount of the corresponding available resource.
Changing the Item Entity Parent
Finally, we remove the item from the market and add it to the hero's inventory. For this, we use the Ecspanse.Command.remove_child!/2 and Ecspanse.Command.add_child!/2 functions.

  
    
  
  The Purchase API


We first need to add the PurchaseMarketItem event to setup as sync system:
#...
|> Ecspanse.add_frame_end_system(Systems.PurchaseMarketItem)
#...
Then expose the event in the API:
@spec purchase_market_item(item_entity_id :: Ecspanse.Entity.id()) :: :ok
def purchase_market_item(item_entity_id) do
  Ecspanse.event({Demo.Events.PurchaseMarketItem, item_entity_id: item_entity_id})
end
Now we can test it in the console. First make sure that the hero has enough resources by walking around and using the exposed fetch_hero_details/0 function. Then check the market items with fetch_market_items/0 and note down the desired item entity ID. Purchase the item with purchase_market_item/1. Finally, check the hero details again to see the item in the inventory.


  
    
  
  Testing the Systems


The goal of this chapter is to learn how to test the systems. We will use the MoveHero system as an example.

	testing systems in isolation
	using a custom Ecspanse setup
	using the system debugger to run systems manually


The game story is now ready. Time to see how we can test it.
Before we start, it is important to note that in test mode, the Ecspanse.Server is not automatically started. This allows us to decide the moment when the server should start, and what systems to run.
There are many ways to test systems. We will choose the most straightforward for this tutorial: testing the systems in isolation. That means that the systems scheduled under Demo.setup/1 are not running. We will create a new setup function for testing, and call the systems manually.
defmodule Demo.Systems.MoveHeroTest do
  use ExUnit.Case, async: false

  defmodule DemoTest do
    use Ecspanse
    @impl true
    def setup(data) do
      data
    end
  end

  setup do
    {:ok, _pid} = start_supervised({DemoTest, :test})
    Ecspanse.System.debug()

    hero_entity = %Ecspanse.Entity{} = Ecspanse.Command.spawn_entity!(Demo.Entities.Hero.new())
    {:ok, position_component} = Demo.Components.Position.fetch(hero_entity)

    assert position_component.x == 0
    assert position_component.y == 0

    {:ok, energy_component} = Demo.Components.Energy.fetch(hero_entity)
    assert energy_component.current == 50

    {:ok, hero_entity: hero_entity, energy_component: energy_component}
  end

  test "hero moves if enough energy", %{hero_entity: hero_entity} do
    event = move(:up)
    frame = frame(event)
    Demo.Systems.MoveHero.run(event, frame)

    {:ok, position_component} = Demo.Components.Position.fetch(hero_entity)
    assert position_component.x == 0
    assert position_component.y == 1

    {:ok, energy_component} = Demo.Components.Energy.fetch(hero_entity)
    assert energy_component.current == 49

    #...
  end

  test "hero doesn not move if not enough energy", %{
    hero_entity: hero_entity,
    energy_component: energy_component
  } do
    Ecspanse.Command.update_component!(energy_component, current: 0)

    event = move(:up)
    frame = frame(event)
    Demo.Systems.MoveHero.run(event, frame)

    {:ok, position_component} = Demo.Components.Position.fetch(hero_entity)
    assert position_component.x == 0
    assert position_component.y == 0

    {:ok, energy_component} = Demo.Components.Energy.fetch(hero_entity)
    assert energy_component.current == 0
  end

  defp move(direction) do
    %Demo.Events.MoveHero{direction: direction, inserted_at: System.os_time()}
  end

  defp frame(event) do
    %Ecspanse.Frame{event_batches: [[event]], delta: 1}
  end
end

  
    
  
  Test Dependencies


We start by creating a DemoTest module and implement a setup/1 function that does not schedule any systems.

  
    
  
  Test Setup


There are 2 things happening at the top of the setup block:
	we manually start the server by passing the {DemoTest, :test} tuple to the start_supervised/1 function. Compared to the normal setup where we add Demo to the supervision tree, the {MODULE, :test} tuple allows us to start the server in test mode.
	we run the Ecspanse.System.debug/0 function. This function "upgrades" the current test PID to a system, which allows us to run systems manually. As mentioned previously, commands can be run only from inside a system. Making the test PID a system allows us to run commands directly in our tests.

For the rest of the setup block we setup the test data. We spawn a hero entity and fetch its position and energy components. We assert that the hero is at the starting position and has the starting energy.

  
    
  
  Test the Hero Can Move


We create two helper functions that would create a MoveHero event and a Ecspanse.Frame.t/0 with the event in it.
Then we can run the MoveHero system manually by simply calling Demo.Systems.MoveHero.run(event, frame).
From there on, we can query any component and do any relevant assertion. In this case, we assert that the hero has moved and that the energy has been reduced.


  
    
  
  Running the Demo


The code for this tutorial, together with instructions on how to run it in Livebook is available on GitHub.
Also, you can find a more complex example of a multiplayer game built with Ecspanse on GitHub.


  

    
Saving and Loading
    

With the introduction of the Ecspanse.Snapshot module, it is now possible to implement custom save and load functionalities. Let's explore some strategies and potential pitfalls.

  
    
  
  Generic Info



  
    
  
  Running Snapshot Functions


The Ecspanse.Snapshot functions should run in synchronous Systems.
Depending on the project needs they may run:
	at startup and shutdown:

def setup(data) do
  data
  |> add_startup_system(Demo.Systems.Load)
  # ...
  |> add_shutdown_system(Demo.Systems.Save)
end
	on demand, in systems that listed to save/load events:

def setup(data) do
  data
  |> add_frame_start_system(Demo.Systems.Load)
  # ...
  |> add_frame_end_system(Demo.Systems.Save)
end
The logic can be further refined with conditional systems (run_in_state | run_not_in_state) and Ecspanse.State.

  
    
  
  Restoring functions


The restoring functions are overwriting the existing entities components or resources. That means if the entity with id "1" and component "A" exists, and a restore function is called for the same entity and component, the component "A" state and tags will be replaced with the one from the restore. Restoring will not affect other existing components of the entity "1" that are not in the scope of the restore.
If the logic requires despawning potentially existing entities before restoring, the following pattern may be used:
case Ecspanse.Query.fetch_entity(entity_id) do
  {:ok, entity} -> Ecspanse.Command.despawn_entity_and_descendants!(entity)
  _ -> :ok
end
Ecspanse.Snapshot.restore_entity!(entity_id, component_specs_list)

  
    
  
  Versioning


The Ecspanse.Snapshot.EntitySnapshot and Ecspanse.Snapshot.ResourceSnapshot structs have a version :: integer() field that can be used to manage backwards compatibility. The version can be used to determine how to restore the entity or resource.
It is the developer's responsibility to provide and update the version field.
defmodule TestServer1 do
  use Ecspanse, fps_limit: 60, version: 1

  def setup(data) do
    # ...
  end
end
Then, upon introduction of a breaking change, the version can be updated: e.g. 2.
For example, we have an entity snapshot with version 3. Meanwhile, the module name for one of the entity components has changed. At this point a 1-to-1 restore will fail, as the old module does not exist anymore. However, being able to check the version, we can intercept the payload and transform it to the new module. More on this in the Custom Save and Load section below.

  
    
  
  Basic 1-to-1 Save and Load


The easiest approach is to use the provided Ecspanse.Snapshot.EntitySnapshot and Ecspanse.Snapshot.ResourceSnapshot structs to save and load entities and resources.
The code below would export all components grouped by entity as a list of Ecspanse.Snapshot.EntitySnapshot structs.
snapshots = Ecspanse.Snapshot.export_entities!()
And this will restore the entities from the list of Ecspanse.Snapshot.EntitySnapshot structs.
Ecspanse.Snapshot.restore_entities_from_snapshots!(snapshots)
That's it!

  
    
  
  Persisting the Snapshots


The persistence mechanism is beyond the scope of the Ecspanse library. This chapter provides some considerations on the topic.

  
    
  
  Encode and Decode


Probably the most common and easiest way to persist data is to encode it to a binary format and save it to a file. It is however important to consider the security implications of this approach. Here are some good explanations of the risks and how to mitigate them.
The :safe options that prevents the creation of new atoms may require special attention. For example if the one encoded component has some atom tags, and the current code does not use those tags anymore, the decoding with :safe option will fail. The same would happen also for any atom in the component state that does not exist anymore in the current code.

  
    
  
  Database Persistence


In essence, the complexity of the project's components directly influences the difficulty of the task. Ecspanse.Component supports any type of data within its state, including deeply nested structs with lists of atoms as fields. This can pose a challenge when modeling in a database.
However, projects with simpler components can be more easily persisted in a database.

  
    
  
  Filtering


Not every component, entity, or resource needs to be saved. Both Ecspanse.Component and Ecspanse.Resource include an export_filter option, which defaults to :none, to filter the data that is exported.
Filtering out a component type from all entities:
defmodule Demo.Components.MyComponent do
  use Ecspanse.Component, export_filter: :component
end
Filtering out entities with a specific component. This is specially useful for components used as tags. For example, in a game happening in a forrest where many leaves that are purely decorative are floating in the background. We decide we do not want to export the leaves but to generate them on demand when the game loads. For this we could add a stateless component Demo.Components.Leaf to the leaves entities and filter the entity out from the export. Even if the leaf entity has more components like Demo.Components.Position or Demo.Components.Velocity, none of them will be exported.
defmodule Demo.Components.Leaf do
  use Ecspanse.Component, export_filter: :entity
end
Filtering out a resource:
defmodule Demo.Resources.MyResource do
  use Ecspanse.Resource, export_filter: :resource
end

  
    
  
  Custom Save and Load for Backwards Compatibility


As highlighted in the Versioning section, exporting and importing data in a 1-to-1 manner is not always feasible. Code and data evolve over time, and saved data may become outdated. In such cases, a custom save and load mechanism is required. Ecspanse.Snapshot provides functions restore entities and resources from Ecspanse.Component.component_spec/0 and Ecspanse.Resource.resource_spec/0 that can be composed by traversing the snapshots.
Let's consider a scenario where we have a Demo.Components.OldComponent that has been renamed to Demo.Components.NewComponent:
# iterating the component specs in an EntitySnapshot
specs = for component_spec <- entity_snapshot.component_specs do
  case component_spec do
    {Demo.Components.OldComponent, state, tags} ->
      {Demo.Components.NewComponent, state, tags}
    _ -> component_spec
  end
end

Ecspanse.Snapshot.restore_entity!(entity_snapshot.id, specs)

  
    
  
  Managing Invalid Entity Relationships


When working with Ecspanse.Command to create inter-entity relationships, the library performs checks to ensure that these relationships are valid. It uses a reflection mechanism to ensure both parent and child entities are aware of each other. Consequently, developers do not need to manually insert, remove, or update Ecspanse.Component.Parent or Ecspanse.Component.Child components.
In contrast, Ecspanse.Snapshot takes a naive approach. The Ecspanse.Component.Parent and Ecspanse.Component.Child components are treated like any other components. Restoring an entity with children does not automatically restore the child entities. Unless these child entities are also included in the list of entities to be restored, the parent entity will contain invalid relationships.
Building upon the previous forest example, let's consider a scenario where leaf entities are children of tree entities. If we decide to export the trees but not the leaves, the trees will be restored without their leaf children. However, the tree entities will still reference the leaves in their Ecspanse.Component.Child component, resulting in an invalid relationship. This won't break the logic but may lead to unexpected behavior or confusion.
To address this issue, developers can manually despawn existing leaf entities or remove their relationship with the trees before exporting. However, this approach may not always be feasible. In such cases, Ecspanse.Snapshot provides two functions to help manage invalid entity relationships:
	Ecspanse.Snapshot.show_invalid_relationships/0 - This function should be run after the restoration process is completed. It will print a list of entities with their invalid relationships.
	Ecspanse.Snapshot.remove_invalid_relationships!/0 - If invalid relationships are expected, this function can be called to remove all of them.



  

    
Changelog
    


  
    
  
  v0.10.0 (2024-09-25)



  
    
  
  Features


	introduces Ecspanse.Snapshot to enable custom save and load functionalities.	use Ecspanse accepts now the :version option to ensure backwards compatibility when restoring entities and resources.
	use Ecspanse.Component and use Ecspanse.Resource accept now the :export_filter option.


	new guides section for save and load.


  
    
  
  v0.9.0 (2024-03-30)



  
    
  
  Breaking


	removes Ecspanse.Resource.State in favor of Ecspanse.State functionality.

Replacing the old Ecspanse.Resource.State with Ecspanse.State
Creating a new state
defmodule Demo.States.Game do
  use Ecspanse.State, states: [:play, :paused], default: :play
end
Multiple states can be defined in the same way:
defmodule Demo.States.Area do
  use Ecspanse.State, states: [:dungeon, :market, :forrest], default: :forrest
end
Conditionally running systems
Old code:
Ecspanse.add_system(
  ecspanse_data,
  Demo.Systems.MoveHero,
  run_in_state: [:play]
)
New code:
Ecspanse.add_system(
  ecspanse_data,
  Demo.Systems.MoveHero,
  run_in_state: {Demo.States.Game, :play}
)
Getting and setting states
Old code:
{:ok, %Ecspanse.Resource.State{value: state}} = Ecspanse.Query.fetch_resource(Ecspanse.Resource.State)
Ecspanse.Command.update_resource!(Ecspanse.Resource.State, value: :paused)
New code:
:play == Demo.States.Game.get_state!()

# Attention! The system running this command must be synchronous.
:ok = Demo.States.Game.set_state!(:paused)
Listening to state changes
defmodule Demo.Systems.OnGamePaused do
  use Ecspanse.System, event_subscriptions: [Ecspanse.Event.StateTransition]

  def run(%Ecspanse.Event.StateTransition{module: Demo.States.Game, previous_state: _, current_state: :paused}) do
    # logic
  end

  def run(_event), do: :ok
end


  
    
  
  Features


	allows inserting resources at startup with Ecspanse.insert_resource/2
	allows state init at startup with Ecspanse.init_state/2
	introduces Ecspanse.State state functionalities. See the breaking changes for more details.
	new library built-in Ecspanse.Event.StateTransition event
	new library built-in Ecspanse.Component.Name component


  
    
  
  Improvements


	Ecspanse.Query.entity_exists?/1 to check if an entity still exists
	Ecspanse.Command.add_and_fetch_component!/2 wrapper to return a component after creation
	Ecspanse.Command.update_and_fetch_component!/2 wrapper to return a component after update


  
    
  
  v0.8.1 (2023-12-22)



  
    
  
  Improvements


	returns an explicit error message when trying to run queries or create events and the Ecspanse server is not running.
	documentation improvements.


  
    
  
  v0.8.0 (2023-11-14)



  
    
  
  Improvements


	refactor the Ecspanse.Projection to include the state of the projection	the projection result is now wraped in a Ecspanse.Projection{} struct, together with the projection state.
	the Ecspanse.Projection.project/1 callback returns now the projection state as well as the projection result.




  
    
  
  Breaking


	the Projection.run?/2 callback has been removed. The functionality is now handled by the Ecspanse.Projection.project/1 callback, by returning :halt.
	the Ecspanse.Projection.project/1 callback should now return also the state of the projection. See the documentation for more details.
	the Ecspanse.Projection.on_change/3 callback takes as second and third argument the Ecspanse.Projection.t/0.
	the Ecspanse.Projection.get!/1 callback now returns a Ecspanse.Projection.t/0.


  
    
  
  v0.7.3 (2023-11-13)



  
    
  
  Improvements


	implement the Projection.run?/2 optional callback to run projections conditionally.


  
    
  
  v0.7.2 (2023-11-05)



  
    
  
  Improvements


	Ecspanse.Projection.on_change/3 is called on Projection server initialization.


  
    
  
  v0.7.1 (2023-10-07)



  
    
  
  Improvements


	Ecspanse.Command.clone_entity!/2 and Ecspanse.Command.deep_clone_entity!/2 now accept an :id option to set the id of the cloned entity.


  
    
  
  v0.7.0 (2023-10-05)



  
    
  
  Breaking


	Ecspanse.Projection.on_change/3 replaces the on_change/2 and now takes both the new projection as well as the previous projection as arguments.


  
    
  
  Improvements


	updating projections after all frame systems have run to return a consistent state.


  
    
  
  v0.6.0 (2023-10-02)



  
    
  
  Features


	introduces Ecspanse.Projection to build state projections across entities and components.


  
    
  
  v0.5.0 (2023-09-28)



  
    
  
  Features


	introduces ancestors queries to query for parents of an entity, the parents of parents, and so on:	Ecspanse.Query.select/2 new option: :for_ancestors_of
	Ecspanse.Query.list_ancestors/1
	Ecspanse.Query.list_tagged_components_for_ancestors/2




  
    
  
  v0.4.0 (2023-09-17)



  
    
  
  Breaking


	removes the automatically generated events: Ecspanse.Event.{ComponentCreated, ComponentUpdated, ComponentDeleted, ResourceCreated, ResourceUpdated, ResourceDeleted}. Use custom emitted events or short-lived components instead.


  
    
  
  Improvements


	improves performance for tagged components. The system loop now runs faster when dealing with tagged components.


  
    
  
  v0.3.1 (2023-08-30)



  
    
  
  Fixes


	fixes a bug where events could be scheduled after they were batched for the current frame, and before the current events are cleared. Causing some events to be lost. Thanks to @andzdroid for identifying and documenting the issue.
	fixes a bug where temporary timers would crash. Thanks to @holykol for finding and fixing the issue.


  
    
  
  Features


	imports Ecspanse.Query and Ecspanse.Command in all systems, so all the queries and commands are available without needing the respective module prefix.
	imports Ecspanse in the setup module that use Ecspanse so the system scheduling functions are available without needing the module prefix.


  
    
  
  v0.3.0 (2023-08-21)



  
    
  
  Features


	adds a new query Ecspanse.Query.list_tags/1 to list a component's tags.
	adds a new query Ecspanse.Query.list_components/1 to list all components of an entity.
	adds a new command Ecspanse.Command.clone_entity!/1 to clone an entity without its relationships.
	adds a new command Ecspanse.Command.deep_clone_entity!/1 to clone an entity with its descendants.


  
    
  
  v0.2.1 (2023-08-20)



  
    
  
  Fixes


	batch all events only by batch_key to avoid race conditions for different events processed by the same system.


  
    
  
  v0.2.0 (2023-08-18)



  
    
  
  Breaking


	use Ecspanse.Component.Timer and use Ecspanse.Event.Timer are now deprecated.
Use use Ecspanse.Template.Component.Timer and use Ecspanse.Template.Event.Timer instead.


  
    
  
  Features


	introduces Ecspanse.Template.Component and Ecspanse.Template.Event to simplify the creation of related components and events.
	adds a new query Ecspanse.Query.fetch_component/2 to fetch a system's component by a list of tags.


  
    
  
  v0.1.2 (2023-08-14)



  
    
  
  Fixes


	removes unneeded dependency plug_crypto
	upgrades dependencies: credo, ex_doc, jason


  
    
  
  v0.1.1 (2023-08-12)



  
    
  
  Fixes


	adds the missing project :package



  

    
Ecspanse.Component.Name 
    



      
A basic name component implemented by the library.
While this component can be used by the game logic, it serves also for debugging purposes
and as a way for third-party libraries to consistently identify entities.

      


      
        Summary


  
    Types

Ecspanse.Event.StateTransition 
    



      
Special library event emitted upon a state change.

  
    
  
  Examples


  %Ecspanse.Event.StateTransition{
    module: Demo.States.Game,
    previous_state: :running,
    current_state: :paused
  }

      


      
        Summary


  
    Types

Ecspanse.Snapshot.EntitySnapshot 
    



      
An entity snapshot.
The :version field is the value provided in the use Ecspanse, version: x at the time of export.
This value is used to ensure compatibility when restoring entities.

      


      
        Summary


  
    Types

Ecspanse.Snapshot.ResourceSnapshot 
    



      
A resource snapshot.
The :version field is the value provided in the use Ecspanse, version: x at the time of export.
This value is used to ensure compatibility when restoring entities.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Ecspanse - ECSpanse v0.10.0
    
    

    


  
  

    
Ecspanse behaviour
    



      
Ecspanse is an Entity Component System (ECS) framework for Elixir.
Note
Ecspanse is not a game engine, but a flexible foundation
for managing state and building logic, offering features like:
	flexible queries with multiple filters
	dynamic bidirectional relationships
	versatile tagging capabilities
	system event subscriptions
	asynchronous system execution


The core structure of the Ecspanse library is:
	Ecspanse: The main module used to configure and interact with the library.
	Ecspanse.Server: The server orchestrates the execution of systems and the storage of components, resources, and events.
	Ecspanse.Entity: A simple struct with an ID, serving as a holder for components.
	Ecspanse.Component: A struct that may hold state information or act as a simple label for an entity.
	Ecspanse.System: Holds the application core logic. Systems run every frame, either synchronously or asynchronously.
	Ecspanse.Resource: Global state storage, similar to components but not tied to a specific entity. Resources can only be created, updated, and deleted by synchronously executed systems.
	Ecspanse.Query: A tool for retrieving entities, components, or resources.
	Ecspanse.Command: A mechanism for changing components and resources state. They can only be triggered from a system.
	Ecspanse.Event: A mechanism for triggering events, which can be listened to by systems. It is the way to communicate externally with the systems.


  
    
  
  Usage


To use Ecspanse, a module needs to be created, invoking use Ecspanse. This implements the Ecspanse behaviour, so the setup/1 callback must be defined. All the systems and their execution order are defined in the setup/1 callback.

  
    
  
  Examples


defmodule TestServer1 do
  use Ecspanse, fps_limit: 60, version: 1

  def setup(data) do
    data
    |> add_startup_system(Demo.Systems.SpawnHero)
    |> add_frame_start_system(Demo.Systems.PurchaseItem)
    |> add_system(Demo.Systems.MoveHero)
    |> add_frame_end_system(Ecspanse.System.Timer)
    |> add_shutdown_system(Demo.Systems.Cleanup)
  end
end
Info
The Ecspanse system scheduling functions are imported by default
for the setup module that use Ecspanse


  
    
  
  Configuration


The following configuration options are available:
	:fps_limit (optional) - integer or :unlimited - the maximum number of frames per second. Defaults to :unlimited.
	:version (optional) - non negative integer - the version of the ECS game logic.
Can be used to determine backwards compatibility when saving and loading state(see Ecspanse.Snapshot for details).
Defaults to 0.

Special Resources
Some special resources, such as FPS, are created by default by the framework.

      


      
        Summary


  
    Callbacks

Ecspanse.Data 
    



      
The Data module defines a struct that holds the state of the Ecspanse initialization process.
This struct is passed to the setup/1 callback, which is used to define the running systems.
After the initialization process the Data struct is not relevant anymore.

      


      
        Summary


  
    Types

Ecspanse.Frame 
    



      
The frame is a struct that encapsulates the state of the current frame.
It holds information such as the time elapsed since the last frame and any batches of events that have been inserted during the previous frame.
This frame struct is available to all systems during the frame.

  
    
  
  Fields


	:event_batches - a collection of event batches queued for execution within this frame.
	:delta - the time elapsed since the last frame in milliseconds.


      


      
        Summary


  
    Types

Ecspanse.Server 
    



      
The server is responsible for managing the internal state of the framework,
scheduling and running the Systems, and batching the Events.
The server is started by adding the module that invokes use Ecspanse to the supervision tree.
Info
There is only one server instance running per otp app.
Trying to create another setup module that use Ecspanse and
adding it to the supervision tree will result in an error.


      


      
        Summary


  
    Types

Ecspanse.Server.State 
    



      
The internal state of the framework.

      


      
        Summary


  
    Types

Ecspanse.TestServer 
    



      
This server is initiated upon application launch when operating in the test environment.
This is done to allow tests to start their own custom servers and schedule custom systems.
A basic test system setup may look like this:
  defmodule Demo.Systems.MoveHeroTest do
    use ExUnit.Case

    defmodule DemoTest do
      @moduledoc "A setup that does not schedule any system"
      use Ecspanse

      @impl true
      def setup(data) do
        data
      end
    end

    setup do
      {:ok, _pid} = start_supervised({DemoTest, :test})
      Ecspanse.System.debug()
    end
  end
  In the test setup block, the server is started with the tuple {DemoTest, :test}.
  This is needed to point Ecspanse to start the Ecspanse.Server with the DemoTest setup.
  The Ecspanse.System.debug/0 call grants the test pid Ecspanse.System powers.
  Meaning that it can runt commands without being in the context of a system.

      


      
        Summary


  
    Functions

Ecspanse.Entity 
    



      
Entities are only identifiers. An entity exists only if it holds at least one component.
The entities per se are not persisted.
Entities are represented as a struct with id as the only field.

  
    
  
  Examples


  %Ecspanse.Entity{id: "cfa1ad89-44b6-4d1f-8590-186354be9158"}

      


      
        Summary


  
    Types

Ecspanse.Component behaviour
    



      
The Ecspanse.Component is the basic building block of the ECS architecture, holding the entity state.
The components are defined by invoking use Ecspanse.Component in their module definition.
An entity cannot exist without at least a component.
And the other way around, a component cannot exist without being allocated to an entity.
The components hold their own state, and can also be tagged for easy grouping.
There are two ways of providing the components with their initial state and tags:
	At compile time, when invoking the use Ecspanse.Component, by providing the :state and :tags options.
defmodule Demo.Components.Position do
 use Ecspanse.Component, state: [x: 3, y: 5], tags: [:map]
end

	At runtime when creating the components from specs: t:Ecspanse.Component.component_spec():
Ecspanse.Command.spawn_entity!({Ecspanse.Entity,
 components: [
   Hero,
   {Demo.Components.Position, [x: 7, y: 2], [:map]},
 ]
)

# or

Ecspanse.Command.add_component!(hero_entity, {Demo.Components.Position, [x: 7, y: 2], [:map]})


After being created, components become structs with the provided fields, along with some metadata added by the framework.
Components can also be used as an entity label, without state.
There are some special components that are created automatically by the framework:
	Ecspanse.Component.Children - holds the list of child entities.
	Ecspanse.Component.Parents - holds the parent entities.


  
    
  
  Options


	:state - a list with all the component state struct keys and their initial values (if any).
For example: [:amount, max_amount: 100]
	:tags - list of atoms that act as tags for the current component. Defaults to [].
	:export_filter - :none | :component | :entity - indicates if the component should be exported.
Defaults to :none. See Ecspanse.Snapshot for details.
	:none - no filter. The component will be exported.
	:component - the component will not be exported.
	:entity - the whole entity containing the component will not be exported.



Tags
Tags can be added at compile time, and at runtime only when creating a new component.
They cannot be edited or removed later on for the existing component.
The List of tags added at compile time is merged with the one provided at run time.


      


      
        Summary


  
    Implemented Callbacks

Ecspanse.Component.Children 
    



      
The Children component is a special component provided by the framework
to maintain references to an entity's child entities.
Dedicated queries and commands are provided to interact with this component.
An empty Children component is automatically added upon entity creation,
even if no child entities are defined at the time of creation.
Ecspanse parent-child relationships are bidirectional associations

Ecspanse.Component.Parents 
    



      
The Parents component is a special component provided by the framework
to maintain references to an entity's parent entities.
Dedicated queries and commands are provided to interact with this component.
An empty Parents component is automatically added upon entity creation,
even if no parent entities are defined at the time of creation.
Ecspanse parent-child relationships are bidirectional associations


      


      
        Summary


  
    Types

Ecspanse.System 
    



      
The system implements the logic and behaviors of the application
by manipulating the state of the components.
The systems are defined by invoking use Ecspanse.System in their module definition.
The system modules must implement the
Ecspanse.System.WithoutEventSubscriptions.run/1 or
Ecspanse.System.WithEventSubscriptions.run/2 callbacks,
depending if the system subscribes to certain events or not.
The return value of the run function is ignored.
The Ecspanse systems run either synchronously or asynchronously,
as scheduled in the Ecspanse.setup/1 callback.
Systems are the sole mechanism through which the state of components can be altered.
Running commands outside of a system is not allowed.
Resources can be created, updated, and deleted only by systems that are executed synchronously.
There are some special systems that are created automatically by the framework:
	Ecspanse.System.CreateStartupResources - startup system that creates the default framework resources, states, and custom resources inserted at startup.
	Ecspanse.System.Debug - used by the debug/0 function.
	Ecspanse.System.Timer - tracks and updates all components using the Ecspanse.Template.Component.Timer template.
	Ecspanse.System.TrackFPS - tracks and updates the Ecspanse.Resource.FPS resource.

Info
The Ecspanse.Query and Ecspanse.Command functions are imported by default
for all modules that use Ecspanse.System


  
    
  
  Options


	:lock_components - a list of component modules
that will be locked for the duration of the system execution.
	:event_subscriptions - a list of event modules that the system subscribes to.


  
    
  
  Component locking


Component locking is required only for async systems to avoid race conditions.
For async systems, any components that are to be modified, created, or deleted,
must be locked in the lock_components option. Otherwise, the operation will fail.
Wherever it makes sense, it is recommended to lock also components that are queried but not modified,
as they could be modified by other systems.
Not all async systems run concurrently. The systems are grouped in batches,
based on the components they lock.

  
    
  
  Event subscriptions


The event subscriptions enables a system to execute solely in response to certain specified events.
The Ecspanse.System.WithEventSubscriptions.run/2 callback is triggered
for every occurrence of an event type to which the system has subscribed.
These callbacks execute concurrently to enhance performance.
However, they are grouped based on their batch keys (see Ecspanse.event/2 options)
as a safeguard against potential race conditions.

  
    
  
  Examples


  defmodule Demo.Systems.Move do
    @moduledoc "An async system locking components, that subscribes to an event"
    use Ecspanse.System,
      lock_components: [Demo.Components.Position],
      event_subscriptions: [Demo.Events.Move]

    def run(%Demo.Events.Move{entity_id: entity_id, direction: direction}, frame) do
      # move logic
    end
  end

  defmodule Demo.Systems.SpawnEnemy do
    @moduledoc "A sync system that does not need to lock components, and it is not subscribed to any events"
    use Ecspanse.System

    def run(frame) do
      # spawn logic
    end
  end

Ecspanse.System.CreateStartupResources 
    



      
Special framework system that creates default resources.
Automatically runs only once on startup.

Ecspanse.System.Debug 
    



      
Generic system to be used by the Ecspanse.System.debug/0
in dev and test environments.

Ecspanse.System.Timer 
    



      
A special system provi